

NAG Fortran Library Routine Document

F08FTF (CUNGTR/ZUNGTR)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08FTF (CUNGTR/ZUNGTR) generates the complex unitary matrix Q , which was determined by F08FSF (CHETRD/ZHETRD) when reducing a Hermitian matrix to tridiagonal form.

2 Specification

```
SUBROUTINE F08FTF(UPLO, N, A, LDA, TAU, WORK, LWORK, INFO)
ENTRY          cungtr (UPLO, N, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER        N, LDA, LWORK, INFO
complex       A(LDA,*), TAU(*), WORK(*)
CHARACTER*1    UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine is intended to be used after a call to F08FSF (CHETRD/ZHETRD), which reduces a complex Hermitian matrix A to real symmetric tridiagonal form T by a unitary similarity transformation: $A = QTQ^H$. F08FSF represents the unitary matrix Q as a product of $n - 1$ elementary reflectors.

This routine may be used to generate Q explicitly as a square matrix.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: UPLO – CHARACTER*1 *Input*
On entry: this **must** be the same parameter UPLO as supplied to F08FSF (CHETRD/ZHETRD).
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix Q .
Constraint: $N \geq 0$.
- 3: A(LDA,*) – ***complex*** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: details of the vectors which define the elementary reflectors, as returned by F08FSF (CHETRD/ZHETRD).
On exit: the n by n unitary matrix Q .

- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08FTF (CUNGTR/ZUNGTR) is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: TAU(*) – **complex** array *Input*
Note: the dimension of the array TAU must be at least $\max(1, N - 1)$.
On entry: further details of the elementary reflectors, as returned by F08FSF (CHETRD/ZHETRD).
- 6: WORK(*) – **complex** array *Workspace*
Note: the dimension of the array WORK must be at least $\max(1, LWORK)$.
On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimum performance.
- 7: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08FTF (CUNGTR/ZUNGTR) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).
Suggested value: for optimum performance LWORK should be at least $(N - 1) \times nb$, where *nb* is the **blocksize**.
Constraint: $LWORK \geq \max(1, N - 1)$ or LWORK = -1.
- 8: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = -*i*, the *i*th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed matrix *Q* differs from an exactly unitary matrix by a matrix *E* such that

$$\|E\|_2 = O(\epsilon),$$

where ϵ is the **machine precision**.

8 Further Comments

The total number of real floating-point operations is approximately $\frac{16}{3}n^3$.

The real analogue of this routine is F08FFF (SORGTR/DORGTR).

9 Example

To compute all the eigenvalues and eigenvectors of the matrix A , where

$$A = \begin{pmatrix} -2.28 + 0.00i & 1.78 - 2.03i & 2.26 + 0.10i & -0.12 + 2.53i \\ 1.78 + 2.03i & -1.12 + 0.00i & 0.01 + 0.43i & -1.07 + 0.86i \\ 2.26 - 0.10i & 0.01 - 0.43i & -0.37 + 0.00i & 2.31 - 0.92i \\ -0.12 - 2.53i & -1.07 - 0.86i & 2.31 + 0.92i & -0.73 + 0.00i \end{pmatrix}.$$

Here A is Hermitian and must first be reduced to tridiagonal form by F08FSF (CHETRD/ZHETRD). The program then calls F08FTF (CUNGTR/ZUNGTR) to form Q , and passes this matrix to F08JSF (CSTEQR/ZSTEQR) which computes the eigenvalues and eigenvectors of A .

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08FTF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA, LWORK, LDZ
PARAMETER       (NMAX=8,LDA=NMAX,LWORK=64*NMAX,LDZ=NMAX)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER        UPLO
*      .. Local Arrays ..
complex         A(LDA,NMAX), TAU(NMAX), WORK(LWORK), Z(LDZ,NMAX)
real            D(NMAX), E(NMAX), RWORK(2*NMAX-2)
CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL         F06TFF, X04DBF, chetrd, csteqr, cungtr
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08FTF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF
*
*      Reduce A to tridiagonal form T = (Q**H)*A*Q
*
      CALL chetrd(UPLO,N,A,LDA,D,E,TAU,WORK,LWORK,INFO)
*
*      Copy A into Z
*
      CALL F06TFF(UPLO,N,N,A,LDA,Z,LDZ)
*
*      Form Q explicitly, storing the result in Z
*
      CALL cungtr(UPLO,N,Z,LDZ,TAU,WORK,LWORK,INFO)
*
*      Calculate all the eigenvalues and eigenvectors of A
*
      CALL csteqr('V',N,D,E,Z,LDZ,RWORK,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.GT.0) THEN
```

```

      WRITE (NOUT,*) 'Failure to converge.'
    ELSE
*
*       Print eigenvalues and eigenvectors
*
      WRITE (NOUT,*) 'Eigenvalues'
      WRITE (NOUT,99999) (D(I),I=1,N)
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL X04DBF('General',' ',N,N,Z,LDZ,'Bracketed','F7.4',
+               'Eigenvectors','Integer',RLABS,'Integer',CLABS,
+               80,0,IFAIL)
*
      END IF
    END IF
    STOP
*
99999 FORMAT (8X,4(F7.4,11X,:))
END

```

9.2 Program Data

F08FTF Example Program Data

```

4                                     :Value of N
'L'                                 :Value of UPLO
(-2.28, 0.00)
( 1.78, 2.03) (-1.12, 0.00)
( 2.26,-0.10) ( 0.01,-0.43) (-0.37, 0.00)
(-0.12,-2.53) (-1.07,-0.86) ( 2.31, 0.92) (-0.73, 0.00) :End of matrix A

```

9.3 Program Results

F08FTF Example Program Results

```

Eigenvalues
      -6.0002          -3.0030          0.5036          3.9996

```

```

Eigenvectors
           1           2           3           4
1  ( 0.7299, 0.0000) (-0.2120, 0.1497) ( 0.1000,-0.3570) ( 0.1991, 0.4720)
2  (-0.1663,-0.2061) ( 0.7307, 0.0000) ( 0.2863,-0.3353) (-0.2467, 0.3751)
3  (-0.4165,-0.1417) (-0.3291, 0.0479) ( 0.6890, 0.0000) ( 0.4468, 0.1466)
4  ( 0.1743, 0.4162) ( 0.5200, 0.1329) ( 0.0662, 0.4347) ( 0.5612, 0.0000)

```
